

Computergestuetzte Mathematik zur Analysis

Lektion 13 (28. Januar)

```
> restart;
```

▼ Gewöhnliche Differentialgleichungen

```
> Dgl := diff(y(x), x) = y(x);
> dsolve(Dgl, y(x));

> # falls Maple eine Loesung findet, so werden alle Loesungen
ausgegeben
> # Integrationskonstante _C1
> AB := y(0) = A;
> dsolve({Dgl, AB}, y(x));
> Dgl := diff(y(x),x,x) + 2*m* diff(y(x),x) + o^2*y(x);
> sol := dsolve(Dgl,y(x));
> sol := rhs(dsolve(Dgl,y(x)));
> Dsys := { diff(w(x),x) + 2*m*w(x) + o^2*y(x)=0, diff(y(x),x) -
w(x) =0 } ;
> syssol :=dsolve(Dsys);
> sol:=rhs(syssol[2]);

> Dgl := diff(y(x),x) - y(x) + x^3 - 3*x + 2 =0;
> Lsg := dsolve(Dgl, y(x));
> eval(Dgl, Lsg);
> # Ueberpruefen der Loesung durch Einsetzen in DGL
> eval(Lsg, x = 0);
> K1 := solve(eval(%, y(0)=1), _C1);
> # Anfangsbedingung gegeben, Konstante ausrechnen
> eval(y(x), {Lsg, _C1=K1});
> # klappt nicht !

> f1 := subs(Lsg, _C1=K1, y(x));
> Lsg2 := dsolve({Dgl, y(0) = 1/2}, y(x));
> f2 := rhs(Lsg2);
> Lsg3 := dsolve({Dgl, y(0) = 3/2}, y(x));
> f3 := rhs(Lsg3);
> # Loesung der verschiedenen AWA zeichnen
> plot([f1, f2, f3], x = -1.5 .. 2, thickness = 2);
```

```

> pl1 := %:
> with(plots):
> # Richtungsfeld der DGL, nach diff(y(x),x) auflösen
> isolate(Dgl, diff(y(x), x));
> v := [1, rhs(%)];
> # Vektorfeld ist tangential an den Graphen jeder Lösung
> pl2 := fieldplot(v, x = -1.5 .. 2, y = -2 .. 5, thickness = 2)
:
> display({pl1, pl2});
> # Lösungen verlaufen "entlang" des Vektorfeldes

```

getrennte Variable

```

> restart:
> Dgl := diff(y(x), x) = exp(y(x)) * sin(x);
> # Lösung nur lokal auf einem Intervall definiert
> Lsg := dsolve(Dgl);
> eval(Dgl, Lsg);
> f := rhs(Lsg);
> eval(f, x=0);
> K1 := solve(% = -1/2, _C1);
> f1 := rhs(eval(Lsg, _C1=K1));
> plot(f1, x = -2 .. 6, thickness = 2);
> # Logarithmus nicht ueberall definiert
> # Argument des Logarithmus untersuchen
> tmp := cos(x)+exp(1/2)-1;
> a := solve(tmp = 0, x);
> evalf(a);
> simplify(subs(x=-a,tmp));
> # Funktion f1 ist nur auf dem Intervall (-a,a) eine Lösung der
AWA

```

mehrere Lösungen einer AWA

```

> restart:
> # Satz von Picard-Lindelöf : AWA unter gewissen
Voraussetzungen lokal eindeutig lösbar
> g := (x^2 + 2*x)*y(x)*exp(y(x)^2)*diff(y(x), x) - 1;
> Lsgn := dsolve({g = 0, y(2) = 0}, y(x));
> # fuer die AWA y(2)=0 gibt es zwei verschiedene Lösungen
> g1 := unapply(rhs(Lsgn[1]), x): g2 := unapply(rhs(Lsgn[2]), x):
> simplify(eval(g,Lsgn[1])), simplify(eval(g,Lsgn[2]));
> g1(2), g2(2);

```

```

> Lsg := dsolve({g = 0, y(2) = 2/5}, y(x));
> # die AWA y(2)=2/5 hat eindeutige Loesung
> f1 := rhs(Lsg);
> simplify(eval(g,Lsg));
> simplify(eval(f1,x=2));
> # Definitionsbereich von f1 bestimmen
> a := solve(f1 = 0, x);
> with(plots):
> # Richtungsfeld der DGL
> v := <1, 1/((x^2 + x)*y*exp(y^2))>;
> # Vektorfeld normieren
> N := sqrt(v[1]^2+v[2]^2);
> w := v/N;

> p := fieldplot(w, x = 1.45..3, y = -0.6..0.6):
> qg := plot([g1, g2], 2..3,colour=[red,green]):
> qf := plot(f1, x=a..3,colour=yellow):
> display({p, qf, qg}, axes = frame);

> # Beispiel mit unendlich vielen Lösungen
> restart:
> Dgl := diff(y(x),x)*sin(x) = 2*y(x)*cos(x);
> AB := y(0) = 0;
> # keine eindeutige Loesung
> Lsg := dsolve({Dgl, AB}, y(x));
> simplify(eval(Dgl,Lsg));
> f := rhs(Lsg);
> plot([seq(subs(_C1=k,f),k=-2..2)],x=-Pi..3*Pi);
> # Maple hat unendlich viele Loesungen gefunden
> # dies sind aber noch nicht alle

> simplify(subs(x=n*Pi,f)) assuming n::integer;
> simplify(subs(x=n*Pi,diff(f,x))) assuming n::integer;
> # man kann den Stellen x=n*Pi differenzierbar von einem
  Loesungszweig auf den anderen wechseln

```